

NAG Library Routine Document

F03ADF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F03ADF calculates the determinant of a complex matrix using an *LU* factorization with partial pivoting.

2 Specification

```
SUBROUTINE F03ADF(A, LDA, N, DETR, DETI, WKSPCE, IFAIL)
INTEGER          LDA, N, IFAIL
double precision DETR, DETI, WKSPCE(N)
complex*16      A(LDA,*)
```

3 Description

F03ADF calculates the determinant of *A* using the *LU* factorization with partial pivoting, $PA = LU$, where *P* is a permutation matrix, *L* is lower triangular and *U* is unit upper triangular. The determinant of *A* is the product of the diagonal elements of *L*, with the correct sign determined by the row interchanges.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

- 1: *A*(LDA,*) – **complex*16** array *Input/Output*
Note: the second dimension of the array *A* must be at least $\max(1, N)$.
On entry: the *n* by *n* matrix *A*.
On exit: *A* is overwritten by the factors *L* and *U*, except that the unit diagonal elements of *U* are not stored.
- 2: LDA – INTEGER *Input*
On entry: the first dimension of the array *A* as declared in the (sub)program from which F03ADF is called.
Constraint: $LDA \geq \max(1, N)$.
- 3: N – INTEGER *Input*
On entry: the dimension of the array WKSPCE as declared in the (sub)program from which F03ADF is called. *n*, the order of the matrix *A*.
Constraint: $N \geq 0$.
- 4: DETR – **double precision** *Output*
- 5: DETI – **double precision** *Output*
On exit: the real part and imaginary parts of the determinant of *A*.
- 6: WKSPCE(N) – **double precision** array *Workspace*

7: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The matrix A is singular, possibly due to rounding errors. The factorization could not be completed. DETR and DETI are set to 0.0.

IFAIL = 2

Overflow. The value of the determinant is too large to be held in the computer.

IFAIL = 3

Underflow. The value of the determinant is too small to be held in the computer.

IFAIL = 4

On entry, $N < 0$,
or $LDA < \max(1, N)$.

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis, see page 107 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03ADF is approximately proportional to n^3 .

9 Example

This example calculates the determinant of the complex matrix

$$\begin{pmatrix} 1 & 1+2i & 2+10i \\ 1+i & 3i & -5+14i \\ 1+i & 5i & -8+20i \end{pmatrix}.$$

9.1 Program Text

```
*      F03ADF Example Program Text
*      Mark 15 Revised. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NMAX, LDA
      PARAMETER        (NMAX=8,LDA=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
```

```

*      .. Local Scalars ..
      DOUBLE PRECISION DETI, DETR
      INTEGER          I, IFAIL, J, N
*      .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX)
      DOUBLE PRECISION WKSPCE(NMAX)
*      .. External Subroutines ..
      EXTERNAL         F03ADF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F03ADF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GE.0 .AND. N.LE.NMAX) THEN
        READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
        IFAIL = 1
*
        CALL F03ADF(A,LDA,N,DETR,DETI,WKSPCE,IFAIL)
*
        IF (IFAIL.EQ.0) THEN
          WRITE (NOUT,99998) DETR, DETI
        ELSE
          WRITE (NOUT,99997) IFAIL
        END IF
      ELSE
        WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,'Value of determinant = (',F9.4,',',F9.4,')')
99997 FORMAT (1X,/1X,' ** F03ADF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

```

F03ADF Example Program Data
  3
  (1.0, 0.0)   (1.0, 2.0)   (2.0,10.0)
  (1.0, 1.0)   (0.0, 3.0)   (-5.0,14.0)
  (1.0, 1.0)   (0.0, 5.0)   (-8.0,20.0)

```

9.3 Program Results

```

F03ADF Example Program Results

Value of determinant = (   1.0000,   0.0000)

```
